TRANSFERS AND UNSTABLE DEGREES IN THE $\mathbb{A}^1$-BROUWER DEGREES PACKAGE FOR MACAULAY2

STEPHANIE ATHERTON, SOMAK DUTTA, JORDY LOPEZ GARCIA, JOEL LOUWSMA, YUYUAN LUO, WERN JUIN GABRIEL ONG, AND RUZHO SAGARAYAJ

Abstract. We describe a significant update to the *Macaulay2* package `A1BrouwerDegrees`. We extend several methods in the previous version of the package to the setting of finite étale algebras, allowing the computation of transfers along finite étale extensions. Additionally, we implement a number of new features for the computation of unstable $\mathbb{A}^1$-Brouwer degrees and manipulation of classes in the unstable Grothendieck–Witt group.

1. Introduction

The $\mathbb{A}^1$-Brouwer degree is a quadratic-form-valued enhancement of the classical Brouwer degree in motivic homotopy theory. Instead of being valued in the integers, the $\mathbb{A}^1$-Brouwer degree (or $\mathbb{A}^1$-degree) of a morphism between affine spaces $f\colon \mathbb{A}^n_k \to \mathbb{A}^n_k$ over a field k for is valued in the Grothendieck–Witt ring $\mathrm{GW}(k)$ of symmetric bilinear forms. Algorithms for computing these degrees and manipulating quadratic forms are provided in *Macaulay2* [8] as part of the package `A1BrouwerDegrees` [2, 3].

We describe an update to this package that improves on its features in several ways:

(1) Computations in the Grothendieck–Witt ring can now be performed over finite étale algebras over fields, extending the previous version of the package which was restricted to $\mathbb{C}, \mathbb{R}, \mathbb{Q}$, and finite fields of characteristic not equal to two.
(2) For a finite étale algebra L over a field k, we implement a function for computing the transfer $\mathrm{GW}(L) \to \mathrm{GW}(k)$.
(3) We implement methods for computing unstable $\mathbb{A}^1$-Brouwer degrees both globally and locally at k-rational points, valued in the unstable Grothendieck–Witt group $\mathrm{GW}^u(k)$.
(4) We implement methods for manipulating unstable Grothendieck–Witt classes over finite étale algebras, including methods for verifying isomorphisms of classes and producing simplified representatives.

1.1. Outline.

In Section 2, we discuss methods for computing with symmetric bilinear forms over a finite étale algebra L over a field k and for computing the transfer $\mathrm{GW}(L) \to \mathrm{GW}(k)$. Section 3 discusses the construction of unstable Grothendieck–Witt classes in *Macaulay2* and methods for manipulating them. Finally, Section 4 discusses methods for computing unstable $\mathbb{A}^1$-Brouwer degrees and the relationship between local and global degrees via a Poincaré–Hopf-type theorem.

1.2. Related work.

The *Macaulay2* package `A1BrouwerDegrees` [2, 3] provides methods for computing local and global $\mathbb{A}^1$-Brouwer degrees for endomorphisms of affine spaces over

2020 *Mathematics Subject Classification.* primary 14F42, 55M25, 68W30; secondary 11E04.

fields and for manipulation of quadratic forms in the Grothendieck–Witt ring $\mathrm{GW}(k)$ of a field k, building on preliminary algorithms of Brazelton–McKean–Pauli [4, 5].

Explicit formulae to compute unstable global $\mathbb{A}^1$-degrees were introduced by Cazanave [6] and further studied by Kass–Wickelgren [10], who compute the $\mathrm{GW}(k)$-factor of the unstable local degree at rational points. Recent work of Igieobo–McKean–Sanchez–Taylor–Wickelgren [9] provides methods for computing local unstable $\mathbb{A}^1$-degrees at rational points and provides a Poincaré–Hopf-type theorem relating local and global unstable $\mathbb{A}^1$-degrees for rational functions with all zeroes k-rational. It is expected that a Poincaré–Hopf-type theorem will hold and be similarly computable in the setting where the zeroes of the rational function are not necessarily k-rational, as conjectured by Adhikari–Hall–McKean [1, Conjecture 4.1].

1.3. **Software availability.** The software described here is available as version 2.0 of the `A1BrouwerDegrees` package in versions 1.25.11 and later of *Macaulay2*. The documentation is hosted on the *Macaulay2* website at the following link:

https://macaulay2.com/doc/Macaulay2/share/doc/Macaulay2/A1BrouwerDegrees/html/index.html

2. Finite étale algebras and Grothendieck–Witt transfers

Let k be a field of characteristic not equal to two. Recall that a k-algebra L is said to be of *finite type* if it admits a surjection from a polynomial algebra over k in finitely many variables [11, Tag 00F3]. A finite-type k-algebra L is said to be *finite étale* if the trace pairing $L \times L \to k$ given by $(x, y) \mapsto \mathrm{Tr}_{L/k}(xy)$ is non-degenerate [11, Tag 0BVH]. Such an algebra is isomorphic to a finite product of finite separable extensions of k [11, Tag 00U3].

2.1. **Computations with finite étale algebras.** Given a finite étale algebra L over a field k, we can define its Grothendieck–Witt ring $\mathrm{GW}(L)$. In the `A1BrouwerDegreesPackage`, elements of Grothendieck–Witt rings have type `GrothendieckWittClass`. Objects of this type can be constructed from their Gram matrices, which are non-degenerate symmetric matrices over L.

```
i1 : needsPackage "A1BrouwerDegrees"
o1 = A1BrouwerDegrees
o1 : Package
i2 : L = QQ[x]/(x^2 - 1);
i3 : M = matrix(L, {{1,2},{2,x}});
o3 : Matrix L^2 <-- L^2
i4 : beta = makeGWClass M
```

$$o4 = \begin{pmatrix} 1 & 2 \\ 2 & x \end{pmatrix}$$

```
o4 : GrothendieckWittClass
```

The underlying Gram matrix and algebra over which a Grothendieck–Witt class is defined can be retrieved by `getMatrix` and `getAlgebra`, respectively. If the class is defined over a field, the field can be extracted using `getBaseField`. The ring operations on $\mathrm{GW}(L)$ are implemented as `addGW` and `multiplyGW`. Since every symmetric matrix is congruent to a diagonal one, every Grothendieck–Witt class admits a diagonal representative. Such representatives can be obtained using `getDiagonalClass`.

```
i5 : getDiagonalClass beta
```

o5 = $\begin{pmatrix} 1 & 0 \\ 0 & x-4 \end{pmatrix}$

```
o5 : GrothendieckWittClass
```

2.2. **Computing transfers.** Given a symmetric bilinear form $\beta\colon L \times L \to L$, composition with the trace $\mathrm{Tr}_{L/k}\colon L \to k$ yields a symmetric bilinear form on k. The *transfer* $\mathrm{Tr}_{L/k}$ is the Abelian group homomorphism $\mathrm{GW}(L) \to \mathrm{GW}(k)$ defined by $\beta \mapsto \mathrm{Tr}_{L/k} \circ \beta$.

We implement a method `transferGW` for computing the transfer of a Grothendieck–Witt class. This relies on our auxiliary methods `getMultiplicationMatrix` and `getTrace` for computing traces of elements of k-algebras more generally, which were not previously implemented in *Macaulay2*.

```
i6 : transferGW beta
```

o6 = $\begin{pmatrix} 2 & 0 \\ 0 & -8 \end{pmatrix}$

```
o6 : GrothendieckWittClass
```

3. The unstable Grothendieck–Witt group

In this section, we let k be a field of characteristic not equal to two and L be a finite étale algebra over k. We define the *unstable Grothendieck–Witt group* $\mathrm{GW}^u(L)$ as the fibered product $\mathrm{GW}(k) \times_{k^\times/(k^\times)^2} L^\times$ in the catetory of Abelian groups under the determinant and quotient maps, respectively. The operation on $\mathrm{GW}^u(L)$ is defined by addition in the first factor and multiplication in the second. Explicit generators and relations for $\mathrm{GW}^u(k)$ are described in [9, Proposition 2.3].

The primary way unstable Grothendieck–Witt classes arise is as pointed homotopy classes of endomorphisms of $\mathbb{P}^1_k$ where k is a field [6, Theorem 3.6]. We further discuss this in Section 4.

3.1. **Constructing unstable Grothendieck–Witt classes.** Our newly introduced type `UnstableGrothendieckWittClass` encodes elements of $\mathrm{GW}^u(L)$ as a `HashTable` consisting of the data of a Gram matrix defined over L and a scalar in $L^\times$. Objects of this type can be constructed via `makeGWuClass`, which takes as input either a pair consisting of a symmetric matrix and scalar defined over the same field or a symmetric matrix over a field. In the latter case, the scalar is assumed to be the determinant of the matrix.

```
i7 : M = matrix(L, {{1,2},{2,x}});
```

o7 : Matrix $L^2 \longleftarrow L^2$

```
i8 : makeGWuClass M
```

o8 = $\left(\begin{pmatrix} 1 & 2 \\ 2 & x \end{pmatrix}, x-4\right)$

```
o8 : UnstableGrothendieckWittClass
i9 : makeGWuClass(M, x - 4)
```

o9 = $\left(\begin{pmatrix} 1 & 2 \\ 2 & x \end{pmatrix}, x-4\right)$

```
o9 : UnstableGrothendieckWittClass
```

Due to the ubiquity of diagonal and hyperbolic forms, we additionally provide constructors `makeDiagonalUnstableForm` and `makeHyperbolicUnstableForm`.

3.2. **Decompositions and isomorphisms in** $\mathrm{GW}^u(k)$**.** In this subsection, we let k be one of $\mathbb{C}, \mathbb{R}, \mathbb{Q}$, or a finite field of characteristic not equal to two. Two fundamental problems in the study of quadratic forms are to determine when two objects of $\mathrm{GW}(k)$ are isomorphic and to produce simplified representatives of a given quadratic form. Algorithms for decomposing quadratic forms over k and verifying isomorphisms are discussed in [3, § 2.1-2] and the references therein. These methods readily generalize to the unstable Grothendieck–Witt group $\mathrm{GW}^u(k)$.

By the structure of $\mathrm{GW}^u(k)$ as a fibered product, two classes in $\mathrm{GW}^u(k)$ are isomorphic if and only if their $\mathrm{GW}(k)$-factors are isomorphic and their $k^\times$-factors are equal. Testing isomorphisms of forms is implemented as `isIsomorphicForm`.

```
i10 : A1 = matrix(QQ, {{1,-2,4},{-2,2,0},{4,0,-7}});
o10 : Matrix QQ^3 <-- QQ^3
i11 : A2 = matrix(QQ, {{1,0,0},{0,-2,0},{0,0,9}});
o11 : Matrix QQ^3 <-- QQ^3
i12 : alpha1 = makeGWuClass A1
```

$$\texttt{o12 =} \left(\begin{pmatrix} 1 & -2 & 4 \\ -2 & 2 & 0 \\ 4 & 0 & -7 \end{pmatrix}, -18 \right)$$

```
o12 : UnstableGrothendieckWittClass
i13 : alpha2 = makeGWuClass A2
```

$$\texttt{o13 =} \left(\begin{pmatrix} 1 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 9 \end{pmatrix}, -18 \right)$$

```
o13 : UnstableGrothendieckWittClass
i14 : isIsomorphicForm(alpha1, alpha2)
o14 = true
```

A class in $\mathrm{GW}^u(k)$ can be simplified by `getSumDecomposition`, which finds a diagonal representative of its $\mathrm{GW}(k)$-factor.

```
i15 : getSumDecomposition alpha1
```

$$\texttt{o15 =} \left(\begin{pmatrix} 2 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{pmatrix}, -18 \right)$$

```
o15 : UnstableGrothendieckWittClass
```

4. Unstable $\mathbb{A}^1$-Brouwer degrees

In this section, we let k be one of $\mathbb{C}, \mathbb{Q}$, or a finite field of characteristic not equal to two. A rational function $f/g\colon \mathbb{P}^1_k \to \mathbb{P}^1_k$ with monic numerator is said to be *pointed* if $(f/g)(\infty) = \infty$. Given a pointed rational function $f/g\colon \mathbb{P}^1_k \to \mathbb{P}^1_k$, work of Cazanave [6, Theorem 3.6] and Igieobo–McKean–Sanchez–Taylor–Wickelgren [9, Lemma 3.10] shows that the global unstable $\mathbb{A}^1$-degree and the local unstable $\mathbb{A}^1$-degree at k-rational points can be explicitly computed in terms of a bilinear form associated to the rational function. We recall these results below.

Remark 1. In general, our techniques for computing these degrees rely on symbolic computation, which in *Macaulay2* can only be performed over exact fields. Over $\mathbb{C}$, we are able to find alternative descriptions of the degrees. We describe these in Remarks 2 and 4.

4.1. **Computing global unstable $\mathbb{A}^1$-degrees.** Let $f/g\colon \mathbb{P}^1_k \to \mathbb{P}^1_k$ be a pointed rational function, where f and g are polynomials in $k[x]$ with no common factors and g not identically zero. The global unstable $\mathbb{A}^1$-degree is given by $(\text{Béz}(f/g), \det \text{Béz}(f/g)) \in \text{GW}^u(k)$ [6, Theorem 3.6], where $\text{Béz}(f/g)$ is the *Bézoutian matrix* $\text{Béz}(f/g) = (a_{ij})_{i,j=0}^m$ and the $a_{ij} \in k$ are such that

$$\frac{f(X)g(Y) - f(Y)g(X)}{X - Y} = \sum_{i,j} a_{i,j} X^i Y^j \in k[X, Y].$$

Remark 2. *Macaulay2* does not allow for the construction of rational functions over inexact fields. However, the discussion in [6, § 3.2] shows that determinant of the Bézoutian matrix can be computed, up to sign, by the resultant of f and g. More specifically, $\det \text{Béz}(f/g) = (-1)^{\frac{n(n-1)}{2}} \cdot \text{Res}(f, g)$, where $n = \deg(f)$. This allows computation of the $k^\times$-factor of a global unstable $\mathbb{A}^1$-degree over inexact fields. In the case of $\mathbb{C}$, the $\text{GW}(k)$-factor is determined by the degree of a rational function over $\mathbb{C}$, which together with the $k^\times$-factor completely determines a class in $\text{GW}^u(\mathbb{C})$.

This is implemented as the `getGlobalUnstableA1Degree` command.

```
i16 : frac QQ[x];
i17 : f = x^5 - 6*x^4 + 11*x^3 - 2*x^2 - 12*x + 8;
i18 : g = x^4 - 5*x^2 + 7*x + 1;
i19 : q = f/g;
i20 : GDq = getGlobalUnstableA1Degree q
```

$$\text{o20} = \left(\begin{pmatrix} -68 & 38 & 11 & -14 & 1 \\ 38 & -63 & 63 & -29 & 7 \\ 11 & 63 & -84 & 39 & -5 \\ -14 & -29 & 39 & -16 & 0 \\ 1 & 7 & -5 & 0 & 1 \end{pmatrix}, -53\,240 \right)$$

```
o20 : UnstableGrothendieckWittClass
```

An alternate version of this command takes in the numerator and denominator of a rational function separately to accommodate the case of rational functions over $\mathbb{C}$. When called on rational functions over $\mathbb{Q}$ or a finite field, the computation is run on the underlying quotient.

```
i21 : CC[x];
i22 : fc = x^5 - 6*x^4 + 11*x^3 - 2*x^2 - 12*x + 8;
i23 : gc = x^4 - 5*x^2 + 7*x + 1;
i24 : getGlobalUnstableA1Degree(fc, gc)
```

$$\text{o24} = \left(\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}, -53\,240 \right)$$

```
o24 : UnstableGrothendieckWittClass
```

4.2. **Computing local unstable $\mathbb{A}^1$-degrees.** For a rational function $f/g\colon \mathbb{P}^1_k \to \mathbb{P}^1_k$ with monic numerator that vanishes at $r \in k$ with order m, the local unstable $\mathbb{A}^1$-degree is given by the pair $(\mathrm{Nwt}_r(f/g), \det \mathrm{Nwt}_r(f/g))$, where $\mathrm{Nwt}_r(f/g)$ is the *local Newton matrix* at r. This matrix may be expressed as the antidiagonal matrix

$$\mathrm{Nwt}_r(f/g) = \begin{bmatrix} 0 & \dots & 0 & a_m \\ \vdots & \iddots & \iddots & 0 \\ 0 & \iddots & \iddots & \vdots \\ a_m & 0 & \dots & 0 \end{bmatrix},$$

where $a_m \in k$ is the coefficient appearing in the Laurent series expansion

$$\frac{g(x)}{f(x)} = \frac{a_m}{(x-r)^m} + \frac{a_{m-1}}{(x-r)^{m-1}} + \cdots + \frac{a_1}{(x-r)} + \text{higher order terms} \tag{1}$$

of g/f at r [9, Lemma 3.10].

The method for finding the values $a_m \in k$ of the Laurent series expansion is not entirely immediate, and we record the following result – likely well-known to experts – for their computation.

Proposition 3. *Let f/g be a pointed rational function. Let $r \in k$ be a root of f of multiplicity m. Write a partial fraction decomposition of g/f as in (1). Then a_m is given by the evaluation of $\frac{(x-r)^m g(x)}{f(x)}$ at r.*

Proof. Writing $F(x) = \frac{(x-r)^m g(x)}{f(x)}$, we have that r is neither a root nor a pole of $F(x)$ since f/g is pointed. That is, F is an analytic function in the sense that it is an element of the stalk $\mathcal{O}_{\mathbb{P}^1_k, r}$ at r. Since r is a k-rational point, it in particular has separable residue field, and we can use the Hasse-derivative analogue of Taylor's theorem [7, Corollary 2.5.14], where the power series expansion of F is given by

$$F(x) = F(r) + \sum_{a \geq 1} D_x^{(a)}(F)(r) \cdot (x-r)^a.$$

Dividing by $(x-r)^m$ once more and comparing coefficients yields the claim. □

The procedure in Proposition 3 is implemented as `getLocalUnstableA1Degree`. Taking `q` to be as in line `i19`, we have:

```
i25 : deg1 = getLocalUnstableA1Degree(q, -1)
```

$$\texttt{o25 =} \left(\left(-\tfrac{5}{27} \right), -\tfrac{5}{27} \right)$$

```
o25 : UnstableGrothendieckWittClass
```

Remark 4. Over $\mathbb{C}$, an adapted version of Proposition 3 can be implemented using numerical algebraic geometry to compute the unstable local $\mathbb{A}^1$-degree.

As in the case of computing global unstable $\mathbb{A}^1$-degrees, there is also a variant that takes in the numerator and denominator separately.

```
i26 : getLocalUnstableA1Degree(f, g, -1)
```

$$o26 = \left(\left(-\frac{5}{27}\right), -\frac{5}{27}\right)$$

```
o26 : UnstableGrothendieckWittClass
```

4.3. **A Poincaré–Hopf theorem at rational points.** The global degree of Section 4.1 and the local degree of Section 4.2 are related by a Poincaré–Hopf type formula. Unlike in the stable setting, where the global $\mathbb{A}^1$-degree is computed as a sum of local $\mathbb{A}^1$-degrees in the Grothendieck–Witt ring $\mathrm{GW}(k)$, the local-to-global principle for unstable $\mathbb{A}^1$-degrees depends on the configuration of zeroes of f/g, that is, on a (Weil) divisor on $\mathbb{P}^1_k$. More precisely, for a rational function f/g all of whose zeroes $r_1, \ldots, r_n$ are k-rational with multiplicities m_i and local unstable degrees (β_i, d_i), the global unstable degree is shown in [9, Theorem 1.1] to be given by the divisorial sum formula

$$\deg^u(f/g) = \left(\bigoplus_{i=1}^{n} \beta_i, \prod_{i=1}^{n} d_i \cdot \prod_{i<j} (r_i - r_j)^{2m_i m_j} \right).$$

Continuing with the example introduced in Sections 4.1 and 4.2, we can compute the local unstable $\mathbb{A}^1$-degree of `q` at its remaining roots $1, 2 \in \mathbb{Q}$.

```
i27 : deg2 = getLocalUnstableA1Degree(q, 1)
```

$$o27 = \left(\left(-2\right), -2\right)$$

```
o27 : UnstableGrothendieckWittClass
i28 : deg3 = getLocalUnstableA1Degree(q, 2)
```

$$o28 = \left(\begin{pmatrix} 0 & 0 & \frac{11}{3} \\ 0 & \frac{11}{3} & 0 \\ \frac{11}{3} & 0 & 0 \end{pmatrix}, -\frac{1331}{27}\right)$$

```
o28 : UnstableGrothendieckWittClass
```

Computing the divisorial sum of the local degrees `deg1`, `deg2`, and `deg3`, this agrees with the global unstable $\mathbb{A}^1$-degree computed in Section 4.1.

```
i29 : degSum = addGWuDivisorial({deg1,deg2,deg3}, {-1,1,2});
i30 : isIsomorphicForm(degSum, GDq)
o30 = true
```

5. Acknowledgments

We warmly thank the organizers of the Madison 2025 *Macaulay2* workshop, supported by NSF DMS-2508868, for providing the opportunity to collaborate on this package update. Sabrina Pauli's vision laid the foundation for this update, and we are especially indebted to her both for this and for several insightful discussions. The third-named author additionally thanks Jonathan Hauenstein for valuable conversations, and the sixth-named author additionally thanks Thomas Brazelton for valuable conversations.

References

[1] S. Adhikari, B. Hall, and S. McKean. Tolerants. Preprint, arXiv:2506.22897 [math.AG], 2025.
[2] N. Borisov, T. Brazelton, F. Espino, T. Hagedorn, Z. Han, J. Lopez Garcia, J. Louwsma, W. J. G. Ong, and A. R. Tawfeek. A1BrouwerDegrees: A1-Brouwer degree computations. Version 1.1. A *Macaulay2* package available at https://github.com/Macaulay2/M2/tree/release-1.25.06-branch#.

[3] N. Borisov, T. Brazelton, F. Espino, T. Hagedorn, Z. Han, J. Lopez Garcia, J. Louwsma, W. J. G. Ong, and A. R. Tawfeek. $\mathbb{A}^1$-Brouwer degrees in Macaulay2. *J. Softw. Algebra Geom.*, 14(1):175–187, 2024.

[4] T. Brazelton, S. McKean, and S. Pauli. a1-degree.sage. https://github.com/shmckean/A1-degree/.

[5] T. Brazelton, S. McKean, and S. Pauli. Bézoutians and the $\mathbb{A}^1$-degree. *Algebra Number Theory*, 17(11):1985–2012, 2023.

[6] C. Cazanave. Algebraic homotopy classes of rational functions. *Ann. Sci. Éc. Norm. Supér. (4)*, 45(4):511–534, 2012.

[7] D. M. Goldschmidt. *Algebraic functions and projective curves*, volume 215 of *Grad. Texts Math.* New York, NY: Springer, 2003.

[8] D. R. Grayson and M. E. Stillman. Macaulay2, a software system for research in algebraic geometry. Available at https://macaulay2.com.

[9] J. Igieobo, S. McKean, S. Sanchez, D. Taylor, and K. Wickelgren. Motivic configurations on the line. *Adv. Math.*, 482(part C):Paper No. 110637, 2025.

[10] J. L. Kass and K. Wickelgren. A classical proof that the algebraic homotopy class of a rational function is the residue pairing. *Linear Algebra Appl.*, 595:157–181, 2020.

[11] The Stacks project authors. The stacks project. https://stacks.math.columbia.edu, 2025.

Stephanie Atherton, Department of Mathematics and Statistics, University of Massachusetts, Lowell, MA 01854, USA

Email address: stephanie_atherton@uml.edu

URL: https://bio.site/toyTeX

Somak Dutta, Department of Mathematics, Texas A&M University, College Station, TX 77843, USA

Email address: somakdutta@tamu.edu

URL: https://somak-dutta.github.io

Jordy Lopez Garcia, Department of Applied and Computational Mathematics and Statistics, University of Notre Dame, Notre Dame, IN 46656, USA

Email address: jlopezga@nd.edu

URL: https://jordylopez27.github.io

Joel Louwsma, Department of Mathematics, Niagara University, Niagara University, NY 14109, USA

Email address: jlouwsma@niagara.edu

URL: https://www.joellouwsma.com

Yuyuan Luo, Department of Mathematics, Princeton University, Princeton, NJ 08544, USA

Email address: luo.yuyuan@princeton.edu

URL: https://yuyuan-luo.github.io

Wern Juin Gabriel Ong, Department of Mathematics, University of Bonn, Bonn, D-53111, Germany

Email address: wgabrielong@uni-bonn.de

URL: https://wgabrielong.github.io

Ruzho Sagarayaj, Department of Mathematics, Texas A&M University, College Station, TX 77843, USA

Email address: ruzhomath@tamu.edu